Building the `series_op` Procedure Interactively

First define the series as in Abramowitz & Stegun

```
> s[1] := 1 + a[1]*x + a[2]*x^2 + a[3]*x^3;
```

$$s_1 := 1 + a_1\,x + a_2\,x^2 + a_3\,x^3$$

```
> s[2] := 1 + b[1]*x + b[2]*x^2 + b[3]*x^3;
```

$$s_2 := 1 + b_1\,x + b_2\,x^2 + b_3\,x^3$$

```
> s[3] := 1 + c[1]*x + c[2]*x^2 + c[3]*x^3;
```

$$s_3 := 1 + c_1\,x + c_2\,x^2 + c_3\,x^3$$

Define a set of unknowns (the coeffs. of s3)

```
> unknowns := {c[1],c[2],c[3]};
```

$$unknowns := \{c_1, c_2, c_3\}$$

Define a 'shorthand' procedure for converting an expression to a polynomial

```
> P := proc(x) convert(x,polynom) end;
```

$$P := \mathbf{proc}(x)\ \mathrm{convert}(x, polynom)\ \mathbf{end}$$

Define a specific series to re-express

```
> series_in := 1 / s[1];
```

$$series\_in := \frac{1}{1 + a_1\,x + a_2\,x^2 + a_3\,x^3}$$

Perform a series expansion to high enough order

```
> series(%,x=0,4);
```

$$1 - a_1\,x + (-a_2 + a_1^{\,2})\,x^2 + (-a_3 + a_1\,a_2 + (a_2 - a_1^{\,2})\,a_1)\,x^3 + \mathrm{O}(x^4)$$

Convert the power series to a polynomial

```
> p1 := P(%);
```

$$p1 := 1 - a_1\,x + (-a_2 + a_1^{\,2})\,x^2 + (-a_3 + a_1\,a_2 + (a_2 - a_1^{\,2})\,a_1)\,x^3$$

Subtract the converted series and `s[3]` (equivalent to equating the series)

```
> s[3] - p1;
```

$$c_1\,x + c_2\,x^2 + c_3\,x^3 + a_1\,x - (-a_2 + a_1^{\,2})\,x^2 - (-a_3 + a_1\,a_2 + (a_2 - a_1^{\,2})\,a_1)\,x^3$$

Extract the coefficients with respect to x

```
> coeffs(%,x);
```

$$c_1 + a_1, c_2 + a_2 - a_1^2, c_3 + a_3 - a_1 a_2 - (a_2 - a_1^2) a_1$$

Covert the coefficient sequence to a set. Order by order, the coefficients must vanish, and Maple assumes "= 0" is there is no "=" in an equation

```
> {%};
```

$$\{c_2 + a_2 - a_1^2, c_3 + a_3 - a_1 a_2 - (a_2 - a_1^2) a_1, c_1 + a_1\}$$

Solve the set of equations for c[1], c[2], c[3]

```
> solve(%,unknowns);
```

$$\{c_2 = -a_2 + a_1^2, c_1 = -a_1, c_3 = -a_3 + 2 a_1 a_2 - a_1^3\}$$

Now read the `series_op` procedure from a plain-text file, and display the procedure definition.

```
> read series4;
```

```
> op(series_op);
```

proc(*series_in*::*anything*)

$$\mathrm{solve}(\{\mathrm{coeffs}(\mathrm{P}(s[3]) - \mathrm{P}(\mathrm{series}(\mathit{series\_in}, x = 0, 5)), x)\}, \mathit{unknowns})$$

end

`series_op` returns a SET of equations which define the coefficients `c[1], c[2], ...` etc. in terms of the `a[i]` and `b[i]`. To extract the value of a specific coefficient, use the subs command.

Here's an example showing how to extract the coefficient `c[4]` for the case `s[3] := 1 / s[1]`

```
> s[1] := 1 + a[1]*x   + a[2]*x^2 +
>              a[3]*x^3 + a[4]*x^4;
```

$$s_1 := 1 + a_1\, x + a_2\, x^2 + a_3\, x^3 + a_4\, x^4$$

```
> subs( series_op( 1 / s[1]), c[4] );
```

$$-a_4 + 2\, a_1\, a_3 + a_2^{\,2} - 3\, a_2\, a_1^{\,2} + a_1^{\,4}$$